Описание URQL для URQ_DOS версии 1.35  от  23.11.2006 года
Автор:  Korwin (Евгений Туголуков)

Данное описание состоит из трех частей: 

1. Введение

2. Учебник

3. Справочник

Полезная информация
URQ (Universal RipSoft Quest) –  обобщенное название нескольких платформ для разработки «консольной» ИЛ (интерактивной литературы) отечественного происхождения, основанных на URQL – соответственно Universal Ripsoft Quest Language.

Известные версии программы-интерпретатора (по материалам WalkyTalky и Korwina):
URQ - интерпретатор, написанный человеком, придумавшим сам URQL - RipOS'ом. URQ запускается в среде Windows (винурка, рипурка). К сожалению, Рип, похоже, прекратил работу над уркой. Последняя стабильная версия - 1.4, а последняя тестируемая - 2.03a. Сайт всех программ РипОса - ripsoft.narod.ru давно не обновляется.
От остальных интерпретаторов URQ отличается возможностью использовать графику и музыку в играх. Нельзя использовать выделение кусков текста цветом, но можно менять внешний вид окна интерпретатора.

URQ_DOS - интерпретатор, работающий в среде DOS (досурка) Создана Виктором Коряновым по следам рипурки. (Развивается, в отличие от винурки, но ме-е-е-дленно... :) Например, мы пинали Вика не меньше года, прежде чем в досурке появилась возможность цветового выделения текста и кнопок, и, помнится, года полтора, прежде чем появились строковые переменные. Но лучше поздно, чем никогда.)
Оперативная страничка досурки - vkoranov.newmail.ru Последняя версия выпущена 30.11.2004, фактически является стандартом языка URQL, поддерживает функции недоступные в URQ, уступает только невозможностью вывести картинки и звук, и необходимостью пользоваться ДОС-русификаторами для текстового ввода через input по-русски (либо ухищрениями со стороны автора квеста, обходящего этот недостаток прямым сканированием клавиатуры, см. процедуру rus_input в Учебнике) 


URQ_WCL – (windows command line) вин-версия (урка-висюль) интерпретатора от Вика Корянова. Странички те же, что и у досурки. Поддерживает звук, но не картинки, проблемы с русским вводом отсутствуют как класс, к тому же имеет меньший размер. Оперативная страничка - vkoranov.newmail.ru
URQ_GUI – сырая вин-версия, работающая в окне. По неизвестным причинам разработка остановлена Виктором Коряновым. Оперативная страничка - vkoranov.newmail.ru

BestURQ 1.0 бета - уж и не помню, кем написано... Написано на qbasic. Выглядит симпатично, но 16-разрядная система поддерживает далеко не все возможности URQL и обеспечивает корректную работу лишь ранних квестов. (Случайно увидел Вик на одном софтовом сайте. Как было сказано автором, работает и на 286 - ДОС-версия. Народ не принял. Досурка от Корянова осталась любимицей. Заслуженно.)

Сайт - besturq.boom.ru
Akurka (акурка)- перспективная разработка от Akela на Visual Basic. Должна поддерживать все возможности досурки и рипурки, переключаясь при необходимости с одного интерфейса на другой. Последняя известная версия на момент написания данного документа AkURQ 1.27 b2, поддерживает html-оформление квестов, математические функции, имеет красивый внешний вид и довольно удобный интерфейс. Разработка продолжается.
OverUrka (Явурка, Оверурка – название пока не устоялось) – интерпретатор URQL на java. Разработчик - OverLord

Разработчик редактора URQ-квестов под названием SMSQ (сам можешь сотворить квест) - Евгений Бычков

Доступность интерпретаторов:  freeware.

Переносимость: Win/9x, Win2000/XP и DOS. Существуют отдельные проблемы совместимости между разными версиями интерпретатора (в частности, версиями под Win и DOS). (Для Win-версии, написанной на VB, также должны быть установлены некоторые нестандартные OCX-компоненты.)

Основные возможности. На основе URQ можно создавать простые квесты, напоминающие книгу с ветвящимся сюжетом. Управление осуществляется не традиционным командным путем, а путем выбора пользователем одного из нескольких доступных вариантов действий (например, нажатием на соответствующую кнопку). Конечно, это существенно ограничивает интерактивность (зато сам процесс игры требует минимум усилий со стороны пользователя).

Программы пишутся на простом Бейсик-подобном языке программирования. Базовые средства языка позволяют (разумеется) выводить текст и управляющие кнопки, выполнять ветвление в зависимости от определенных условий, вызывать подпрограммы. В последних версиях допускается шифрование кода.

Ссылки:

http://ripsoft.narod.ru/Ripsoft.htm: RipSoft, старый официальный сайт.

http://urq.fastbb.ru/ - форум по URQ

http://www.urq.allquests.ru:  - официальный сайт по URQ

http://urq.allquests.ru/download/quests.shtml  - большой выбор URQ-квестов.

http://vkoranov.newmail.ru- страничка, откуда можно скачать последнюю версию urq_dos.exe - наиболее популярного интерпретатора urq-квестов

ВВЕДЕНИЕ

Давным-давно, когда скорость работы компьютеров измерялась килогерцами (даже не десятками килогерц!), а оперативная память байтами и, если повезет, килобайтами, (внешней памятью тогда служили громоздкие шкафы цифровых магнитофонов), так вот, в те незапамятные времена программисты тоже играли в игры! Самыми первыми играми были адвентюры (текстовые квесты), которые появились, похоже в 60-е – 70-е годы, когда еще и не пахло никакой графикой, и даже вывод иногда осуществлялся не на дисплей монитора, а на принтер! Самое удивительное заключается в том, что по мере развития технологий интерес к текстовым играм не только не сократился, а вырос, возросло и количество энтузиастов IF – Interactive Fiction (интерактивной литературы). Просто на фоне громадного числа любителей 3D-шутеров и стратегий сравнительно малочисленные интеллигентные любители квестов не так заметны. Если Вы уже играли в текстовые квесты и хотите попробовать себя в качестве их разработчика, можете читать дальше. Если же Вы хотите узнать побольше об адвентюрах и квестах – отошлю вас к великолепной статье С.Симоновича  «Уроки Адвентюры» и другим хорошим статьям на www.taplap.ru и www.urq.allquests.ru.

Назначение программы

URQ - несложная в освоении платформа для проигрывания текстовых квестов преимущественно с менюшным вводом, то есть через систему меню и кнопок. Программа urq_dos.exe представляет собой интерпретатор файлов квестов: она проигрывает файлы .qst(текстовые) и .qs1, .qs2 (шифрованные) формат которых (язык Universal Ripsoft Quest Language, - далее URQL) и будет обсуждаться в дальнейшем.

В текущей версии URQ_DOS (Далее urq_dos или просто "досурка"), как и в Win URQ(от RipOs`a - в дальнейшем просто Win URQ или "винурка") нет встроенного механизма создания квестов - их надо будет писать самому, в каком либо текстовом редакторе или в специально созданном для этого редакторе SMSQ.
Благодарности:

Прежде всего, хотелось бы поблагодарить RipOs`a за создание его программы, его вклад в дело развития уркунизма навсегда отмечен буквой 'R' в названии интерпретатора; во-вторых, Виктора Корянова, за труд его и вдохновение, позволившие нам работать с дос-версией URQ; в третьих, всех авторов замечательных квестов, которые научили меня, - на примере своих творений, - создавать свои произведения, и, наконец, в четвертых, всех авторов документаций по URQL, из которых я беззастенчиво и нагло брал понравившиеся мне фрагменты. Меня лишь отчасти может извинить то, что я честно, (в отличие от авторов многих рефератов, курсовых и дипломных работ, диссертаций и монографий) признаюсь в этом грехе! Кроме того, утверждения авторов документаций были проверены на практике, и исправлены в тех местах, где это было необходимо для приведения в соответствие с текущей версией программы. Я упомяну этих авторов и, надеюсь также, что слово RipOs`a, Etev`a, В.Корянова и WalkyTalky не будет забыто отчасти благодаря мне...

Отдельно благодарю: 

Евга - за ценные консультации по досурке и создание SMSQ-редактора;

Krol'a - за новаторские идеи и опыты по созданию объектно-ориентированного кода на URQL;

Terracon'a - за неутомимое создание огромного квеста TRION, вдохновляющее меня на работу, и примеры эффективного кода;

WalkyTalky - за то, что он скромный и просто умница;

Fighter'a - за его умные вопросы, которые многое проясняли.

Byte'у - спасибо за знакомство с QSP, после которого я взглянул на URQ по-другому;

GrAnd'у - спасибо за сотрудничество при изучении RTADS, - лучшей на сегодняшний день русскоязычной платформы для создания текстовых адвентюр;

Стасу Старкову - низко кланяюсь за высокую требовательность к моим квестам, благодаря которой они получились неплохими.
Akela – за то, что он не «флудер», а трудоголик и делает то, что нужно всем – новую версию URQ!
И всех тех, кого я не назвал, но всё равно помнил, создавая этот документ, - тоже благодарю.

Краткая история URQL

URQ была cоздана вскоре после одной из первых известных в Рунете русских программ для написания квестов OrcZero примерно в 1999 году, автор URQ скрывается под псевдонимом RipOs. Первоначально URQ была сделана на Visual Basic, и предназначалась для создания книг-игр и простых менюшных квестов, то есть предоставляла возможность описать локацию и, по нажатию кнопки выбора, перейти на другую локацию. Была возможность работы с инвентарем – числовыми переменными. Простота платформы привлекла внимание создателей квестов, первой серьезной разработкой  по воспоминаниям тех, кто присутствовал при рождении URQ  стал "Древний кинжал" - большой и интересный квест от Кащея, написанный по мотивам произведений Д.Р.Р.Толкиена. Другими ранними квестами, очаровавшими игроков, были коротенькие "Приключения хомяка Семена" в двух частях (третья появилась намного позднее и была написна уже другим автором). Чуть позже Кащей выпускает вторую часть "Древнего кинжала", которая, как и первая, отличалась от прочих квестов трудностью прохождения в связи с многочисленными вариантами смерти главного героя. Да,  кроме  Кинжалов,  в  начале был еще Странник - по содержанию квест объективно  слабоватый, но там Смайлик впервые показал приемы, которые потом  использовались  во многих разработках того времени. Эпохальным событием был выход демо-версии очень большого квеста от Etev'a, - "Без права на надежду", который многие до сих пор, - и не без оснований, - считают лучшим квестом на URQ. Это был квест с использованием практически всех тогдашних мультимедийных возможностей Win URQ: музыки, картинок, - и при этом отличался очень хорошим литературным стилем, в сравнении с прочими "Бананами" и "Петровичами".  Платформа развивалась, RipOs вносил изменения и новые возможности, дошел до версии 2.0 альфа 3, - очень интересной, но совершенно неудовлетворительной в плане устойчивости в работе, - и внезапно прекратил работу над URQ. Последняя относительно устойчиво работающая версия - 1.4 позволяет вставлять в квест картинки (в 6 разных форматах) и отслеживать клики мышкой по этим картинкам, использовать звуки и музыку (wav  и mid), делать квесты, состоящие из нескольких файлов, и многое другое, в том числе ввод чисел и строк пользователем (но это еще не парсер). Параллельно RipOsу разработку урки вел серьезный программист, скрывающийся под именем Виктор Корянов :). Он писал (и продолжает писать) на Си Dos-версию URQ, которая так и называется URQ_DOS. Вот что он сам пишет о создании досурки:

"Почти все здесь, наверное, знают как я пришел к Урке, но поскольку соответствующие посты и текстовые файлы имеют обыкновение теряться и закапываться :) ... 

Это был самый конец 2000 года, буквально несколько дней до праздника. Так получилось, что у меня в то время на «домашней» машине не было хоть сколько-нибудь нормального монитора, приходилось как-то жить на переделанной советской Электронике (модель уже не помню). Более или менее приемлемо этот агрегат работал только в текстовом монохромном режиме EGA, графический EGA теоретически был, но постоянно в нем находиться было трудно из-за проблем с частотой - быстро уставали глаза (теоретически был и VGA, но только для быстрого просмотра графики - что-то читать в нем было невозможно совсем). Жил я, в общем, полностью в мире текстового Доса - много читал с экрана в Дос Навигаторе, играл в Адом, программил в bc++ 3.1 ... Естественно, иногда искал в Инете на университетской машине что-нибудь соответствующее и как-то раз по запросу «текстовые квесты» (или, может, «текстовые игры») нашел сайт Кащея (текстовыми играми интересуюсь давно, еще с до-PCшных времен). Там я сразу обратил внимание на Древний кинжал, Урка Рипа не запустилась из-за традиционного для старых версий Windows отсутствия ВБ-библиотек, да даже если и запустилась бы, все равно я тогда грузил его только в случае совсем уж крайней необходимости (кстати, в 95-ом нет штатной поддержки EGA). Самая первая версия Досурки была написана буквально в Новый Год за два дня."

Так что, если бы у В.Корянова тогда под рукой была монитор получше, мы бы, возможно, и не увидели этого великого творения, которое обеспечило URQ-квестам распространенность, популярность и  дальнейшее развитие. А может быть, и увидели бы, так как текстовыми квестами он интересовался давно, и, как говорили еще древние римляне, "истина побеждает всё"! 

Ввиду того, что Досурка была:

а) гораздо устойчивее и быстрее в работе;  

б) требовала меньше ресурсов; 

в) имела интерфейс, до ностальгической боли напоминавший старую добрую DOS (серые буквы на черном фоне - наследие монохромного монитора В.Корянова), под которой работали классические текстовые квесты 80-х годов; 

г) имела полезные функции и свойства, отсутствующие в URQ-Win, 

- она получила большее распространение и в настоящее время является стандартным проигрывателем urq-квестов.

В версии 1.22 даже была реализована возможность создания exe-файла квеста, к сожалению, утерянная в последующих версиях (будем надеяться, что автор платформы вернется к этой возможности). 

Комментарий разработчика:

"Тут  есть свои "за" и "против". Доводы "за" вполне очевидны - exe'шник удобнее  распространять  на  просторах  общей Сети (например, выложить свой  квест на сервер типа download.ru) и фиксированная версия Досурки страхует от возможных будущих проблем с совместимостью из-за изменений в  языке,  но  и  доводы  "против" достаточно веские - все улучшения в интерфейсе  и  возможный  выход  нормального  GUI-варианта напрямую не коснутся   уже   скомпилированного  exe'шника  ...  Когда  возможность линковки  работала  для текущих версий Досурки, ей пользовались крайне редко - чуть ли ни один-два раза за все время."

Квесты пишутся  в любом текстовом редакторе, - что также очень приятно, - хотя существует специальный текстовый редактор SMSQ, расшифровывающийся, как "Сам Можешь Сотворить Квест". Он первоначально предназначался для написания квестов для проигрывания на сотовых телефонах, то есть SMS-игр. Редактор сделан Евгением Бычковым (партийные клички "Евг" и "brevno"), и очень облегчает работу с большими квестами. Встроенная в версию 8 редактора возможность определить СК (вероятно, суммарный коэффициент?) квеста на некоторое время вызвала ажиотаж и гонку среди авторов квестов в попытках достичь максимального рейтинга. После появления ТРИОН’а – гигантского квеста от Terracon’a который набрал СК 20+++ ажиотаж утих :).
Идеология URQL.

Достоинства языка URQL (Universal Ripsoft Quest Language) для новичка:  

1. Простота освоения. Достаточно выучить знак метки-начала локации и 3 оператора (END - конец локации, РLN - вывод текста и BTN – кнопка перехода), чтобы уже сделать книгу-игру. Смотри пример в учебнике.

2. Неплохая документация, которую можно найти на сайтах создателей URQ. Данный текст тоже является документацией, причем, по моим данным, самой полной.

3. Огромное количество примеров в виде уже созданных квестов. Все ссылки не привожу, ибо их очень много. Основное собрание на www.urq.allquests.ru, оттуда можно перейти на другие сайты. Квесты могут быть зашифрованы от просмотра, но обычно это не делается - кому же интересно смотреть сначала внутрь игры, а потом играть - всё удовольствие пропадет. Таким образом, обычная практика такова: сначала в квест играют, потом смотрят, как он устроен внутри. Потом делают свой. Благодаря этой "открытости кода" количество квестов на URQ превосходит все другие русские платформы вместе взятые. Смею надеяться, что так оно будет и далее.

4. Эта платформа – «живая», то есть на ней уже написаны квесты, она продолжает развиваться, активно обсуждаться и т.д. Следовательно, пока она не собирается тихо умереть в забвении, подобно десятку других платформ. Кстати, к числу "живых" в настоящее время относятся, по моему мнению: URQ, QSP - менюшные, RTADS, Adrift, RINFORM - с текстовым вводом команд. Если я кого-то забыл - поправьте меня. Новичку легко получить консультацию по любому вопросу у мэтров на форуме www.urq.fastbb.ru или у многочисленных сторонников этой платформы. Для других «живых» платформ это, разумеется, также возможно - на www.taplap.ru и соответствующих сайтах. (QSP - www.qsp.org.ru, RTADS - www.rtads.org, и т.д.)
5. URQ имеет богатые возможности для создания квестов*, а создание квеста требует НА ПОРЯДОК меньше времени, чем создание текстовой адвентюры. Это ОЧЕНЬ ВАЖНО, особенно для начинающих «квестописателей», которые таким образом получают возможность начать с небольших, но вполне работоспособных квестов, а потом подняться к вершинам мастерства.

Глоссарий

*Текстовые квесты - это жанр менюшных игр,  в отличие от текстовых адвентюр, имеющих парсер - синтаксический анализатор для распознавания введенных пользователем с клавиатуры команд.

Достоинства языка URQL (Universal Ripsoft Quest Language) для профессионала:

1. Работа с предметами инвентаря (путем некоторых ухищрений, их даже можно выложить в любом месте, снова взять или передать другому персонажу) - любителей систем RTADS, Adrift, RINFORM - прошу не смеяться. Мы создаем этот механизм сами и там, где это нам нужно. А вы пользуетесь уже готовым. Это не упрек, а просто констатация факта;

2. Наличие переменных (числовых и строковых, в том числе, системных), позволяет создавать как угодно много флагов, счетчиков;

3. Управление цветом фона и текста, причем даже в одной строке;

4. Возможность ввода числа или строки пользователем;

5. Наличие локации common, исполняющейся автоматически при переходах из локации в локацию и позволяющей организовать смену дня и ночи, подсчет ходов, голод, "запалы", "демоны" и т.п.;

6. Подстановка значения переменной в код квеста во время исполнения программы (!), обеспечивающее невероятную гибкость по сравнению с другими платформами. Этот специфический инструмент урки требует определенной привычки, но зато позволяет с легкостью манипулировать не только выводом текста на экран, но и логикой выполнения самой программы, например, организовать массивы, добавить во время выполнения новые кнопки или переименовать их как угодно, осуществить арифметические и строковые функции и т.д.;

7. С 2003 года urq_dos.exe является 32-битным приложением, что обеспечивает довольно быструю работу с квестами*;

8. В последней версии появились функции работы со строками, благодаря которым на досурке уже можно сделать простой парсер(!) (синтаксический анализатор);

9. Процедуры позволяют сократить размер квеста при многократном использовании одного и того же фрагмента;

10. Конструкция if then else и логические операторы not, or, and позволяют несложно запрограммировать сколь угодно сложные условия в квесте и реализовать циклы;

11. Реализована, хотя пока еще не слишком совершенно, поддержка музыки (wcl-версия);

12. Отсутствие ограничений на размер квеста, количество переменных, персонажей, локаций и т.д. Если быть точным, длина строковой переменной в досурке может достигать 2 Гб, количество локаций измеряться сотнями тысяч, количество переменных ограничено только размером виртуальной памяти Вашего компьютера…

13. Скорость вывода текста на экран можно задавать самому в самом квесте - от медленного построчного, - а при помощи оператора pause хоть побуквенного! - вывода в напряженных моментах, до, пусть примитивной, анимации;

14. Появилась возможность использования нескольких действий в одной кнопке – xbtn.

Разработка продолжается, так что если Вам чего-то не хватает в работе, можно обратиться к автору интерпретатора - и, вполне возможно, недостающая функция появится.
*Комментарий разработчика:

"Основное  простое  ощутимое преимущество 32-х разрядного мира – легкая доступность  по  умолчанию  всей  памяти  машины  - RAM и виртуальной. 16и-битная  версия работала только в пределах 640 Kb Доса, так что там были  заморочки с допустимым размером квеста и сейвов (сейвы не больше 64  Kb),  теоретически  могли  быть  проблемы с нехваткой памяти когда переменных  и  меток  очень  много и т.п. На счет скорости - да, - 32-х разрядные  приложения  быстрее выполняются на современных процессорах,

но тут разница не такая критическая и ощутимая. Еще длинные имена - в 16-битной версии не было их поддержки."

Недостатки URQ: 

1. Это НЕ ОБЪЕКТНО-ОРИЕНТИРОВАННАЯ ПЛАТФОРМА.  (Впрочем, профи, - поклон Krol`у, - могут писать структурные программы на неструктурных языках и даже самостоятельно создавать классы и объекты на языках программирования к этому изначально не предназначенных.) Поэтому создание по настоящему БОЛЬШИХ ИГР на URQ непросто, так как требует значительной самодисциплины, чтобы писать комментарии, необходимые для предотвращения путаницы в голове! 

2. Профессиональное программирование на урке требует опыта и терпения (интересно, существует ли хоть один язык программирования, у которого нет этого недостатка?).

3. Windows-версия URQ в настоящее время уступает ее DOS-версии по всем параметрам (возможно, с появлением акурки или URQ_GUI это будет исправлено), за исключением возможности работы с картинками, и при этом существует проблема их совместимости,  то есть, квесты, понимаемые досуркой, могут не работать на Win URQ. Правда, сама досурка, без проблем понимает все квесты написанные для ее Windows-сестрички, только не выводит пока картинки и, соответственно, не поддерживает операций с ними; не понимает, - увы, - настроек интерфейса для Windows; и требует при проигрывании некоторых квестов, использующих переменные-счетчики посещений локаций ввода специального ключа при запуске.

4. Иногда возникают проблемы при вводе русского текста под разными версиями Windows в URQ_DOS32-версии. Они преодолимы при запуске программы из под FAR с внешним русификатором, или же использовании urq_dos.exe wcl-версии (которая, похоже, сейчас становится основной, WCL - Windows Command Line).

5. Написать на URQL не примитивный парсер типа «глагол+существительное», но полноценный синтаксический анализатор (с учётом падежей и большим словарём синонимов) в настоящее время настолько сложно и тяжело, что возможно только теоретически. Если вы сторонник текстовых адвентюр с чисто текстовым вводом комманд, советую использовать прекрасные программы RTADS, Adrift, RINFORM подробную информацию о которых можно получить на сайте www.taplap.ru. Через пару недель освоения вы сможете создавать на них учебные квесты из 5-10 локаций (на URQL - за два дня), а за полгода трудов сделаете что-нибудь действительно интересное, красивое и, возможно, вечное.

А если Вы хотите писать квесты легко и просто - милости прошу к нам, на www.urq.ru и используйте замечательную программу URQ_DOS.EXE.

УЧЕБНИК

Предисловие к учебнику.

 Язык URQL несложен для освоения даже тех авторов квестов, которые не имеют никаких навыков в программировании. Поверьте мне, освоить русский язык гораздо сложнее!  А это руководство  без большого труда позволит вам овладеть всеми хитростями квестописания, самым логичным и естественным путем - от простого к сложному. Итак, начнем.

В любом текстовом редакторе, - я, например, в обычном Notepad'е часто пишу, - создайте в меню Файл(File) новый файл, и в самом начале напишите

; Это будет мой первый квест

То что вы написали - комментарий, который игнорируется интерпретатором и служит в качестве напоминания или пояснения автору квеста, а также тем, кто этот квест потом будет просматривать в учебных целях.

Нажмите Enter и перейдите на следующую строку. Дело в том, что комментарий после ; продолжается до конца строки, а конец строки это такой невидимый символ, который вводится как раз нажатием клавиши  Enter. На новой строке поставьте двоеточие и без всякого пробела за двоеточием напишите название локации, например вот так

:начало   ; Двоеточие+название это метка. 

Интерпретатор начинает работу с первой метки, после нее он ищет и выполняет операторы, пока не встретит оператор останова end. Встретив end интерпретатор выводит на экран все встреченные ранее кнопки и приостанавливает работу, ожидая команды пользователя. Объяснить всё это несколько труднее, чем показать, поэтому просто напишите пример, который я вам приведу ниже и сохраните ваш квест под именем L.qst. Затем запустите его на выполнение из командной строки urq_dos.exe L.qst и посмотрите, что получилось.

Соглашения

Это руководство написано на основе URQ_DOS версия 1.35 от 30.11.2004 года. Поэтому применение его к более ранним версиям, да и использование их, не имеет смысла, а возможности более поздних версий, наверное, будут несколько шире. Однако, к чести разработчика, В.Корянов прилагает просто героические усилия по сохранению совместимости программ снизу, поэтому, смею надеяться, то, что здесь написано, останется достоверным и для поздних версий. 

В документации встречаются специальные символы - квадратные скобки - то, что содержится между ними, - необязательная часть.

Пример: [Имя]Фамилия можно писать как ИмяФамилия, так и Фамилия. Если опустить Имя, то вместо него будет подставлено значение по умолчанию. 

После такого знака [new!] идут нововведения, которых не было до июля 2004 года.

Общие сведения.

urq_dos.exe - интерпретатор квестов, которые существуют в виде текстового файла в windows(CP 1251)- или dos(866)- кодировке и имеют расширение qst, qs1, qs2. Первое расширение qst - просто текстовый файл, второе и третье - зашифрованные т.е. закрытые от просмотра квесты. .qs1-шифровка как у RipOs`a, .qs2 - шифровка В.Корянова (более надежная). В отличие от Win URQ, шифрованные квесты работают в urq_dos с такой же скоростью, как и не шифрованные. (Способ шифровки см. главу Шифрование квеста.)

Urq_dos.exe существует в настоящее время в двух равноправных вариантах: dos32-версия  и wcl-версия, которые отличаются во-первых размером (первая около 64 кб, вторая около 20 кб), во-вторых, скоростью работы на различных версиях Windows (первая, по наблюдениям пользователей шустрее на Win9x, а вторая на Win2000 – XP). И та, и другая программа, тем не менее, работает на любой версии Windows от 95-й до Windows XP (на 3.11 не тестировалась). При запуске wcl-версии под Win2000, WinXP очень рекомендуется заглянуть по правому клику мышки в свойства окна и установить там количество строк на экране 25 вместо 300. Длина квеста до 2 Гб, применяется три вида переменных: переменная инвентаря – вещественная, беззнаковая (то есть не может быть меньше нуля), более того, если равна нулю – исчезает; числовая переменная real long от – 2 до +2; и строковая длиной до 2 Гб. Как уже говорилось выше, размер квеста практически ограничен лишь используемой виртуальной памятью. 

Для создания квестов может использоваться любой тестовый редактор, позволяющий работать с файлами формата .txt в кодировке Windows(CP-1251) или DOS-866. Самым распространенным из них считается редактор notepad (блокнот), поставляющийся вместе с операционной системой Microsoft Windows.

Созданный текстовый файл сохраняется с расширением qst, например my.qst кладется в папку с программой urq_dos.exe и запускается из командной строки командой urq_dos.exe my.qst (в самом простом варианте). Возможно использование бат-файла. Формат запуска программы следующий:

     urq_dos [-acrdl] <путь к qst-файлу>

     Ключи:

    -a     использовать кодировку DOS. Используйте,  если  запускаемый qst-файл написан в кодировке DOS.  Работа  с  save-файлом в этом режиме происходит в этой же кодировке.

    -c     загружаемый qst-файл зашифрован методом из URQ 1.22 . Для  создания  зашифрованных  qst-файлов  можно использовать либо непосредственно URQ, либо утилиту urq_code.

    -r     восстановить игру из qsv-файла (имя отличается от имени qst-файла только расширением)

           Комментарий разработчика:  "Опыт  показывает,  что в существующих квестах для URQ редко                  возникает необходимость  в  более  чем  одном  qsv-файле,  поэтому задача поддержки qsv-файлов с произвольными  именами имеет для меня довольно низкий приоритет по сравнению с другими           направлениями доработок. Пока  что  qsv-файл,  с  которым  работает программа, должен иметь  тоже имя, что и квест, но отличаться от него расширением .qsv  (например,  если  квест  -  stranger.qst,  то  qsv-файл -           stranger.qsv).  Запись  по ходу игры будет производиться в него же."

     -d       Режим  отладки:  интерпретатор  выдает при каждом ходе массу информации о состояни переменных, сделанных действиях и т.д.

     -l       В  этом  режиме urq_dos.exe  при  прохождении квеста создает лог-файл игры с тем же именем, что и у квеста, но с расширением 000,  001  и т.д.  Весьма  полезный  режим,  как для сохранения истории Вашей  героической борьбы с загадками квеста, так и для поиска орфографических ошибок - если  загрузить лог-файл в Word и проверить орфографию, многие ляпы,  опечатки и грамматические ошибки будут легко обнаружены.

  -nocp    В этом режиме переменные-счетчики 'count_имя-локации' переименовываются в просто имя_локации' (для совместимости с Win URQ).

Игра

     Во  время  игры  Ваши  действия  аналогичны  действиям, которые бы Вы

  совершали, играя под URQ для Windows. Окно информации здесь представляет

  из  себя  весь  экран,  а  кнопки  -  пункты меню, один из которых нужно

  выбрать для совершения перехода. Выбор осуществляется курсорными клавишами

  на клавиатуре и Enter.

     Кроме стрелок управления курсором доступны следующие клавиши:

     i     от inventory - посмотреть инвентарь

              Просмотр инвентаря реализован в виде системной локации.

     u     от use - использовать инвентарь. Если в квесте предусмотрены

           соответствующие USE-локации, будут доступны действия с предметами.   

     `     выход в DOS
              На  клавиатуре  эта  клавиша  находится  непосредственно под

           клавишей ESC. После ее нажатия программа немедленно прекратит

           свою работу, без дополнительных вопросов и подтверждений.

     ESC   выход из игры

              Текущая  игра  прекращается  и  игроку  предоставляется  три

           возможности:  начать  игру с начала, загрузить ее из qsv-файла,

           или выйти из URQ_DOS. В  случае, если очередная локация не содержит             кнопок (как правило это означает  смерть  игрока),  программа  ведет            себя  так  же,  как  и при обработке клавиши ESC.

      a    Полностью "рассекречены" средства отладки в связи с активным

           тестированием программы. В режиме просмотра диагностики можно

           проследить за работой двух анализаторов, и, если вдумчиво 

           проанализировать эту информацию, то можно разобраться в причинах 

           практически любого случая некорректной работы Урки. Включается пока 

           просто клавишей a - для удобства - мне сейчас самому часто нужно 

           переключать этот режим. 

     *Режим разработчика:

     Чтобы   активизировать   режим   разработчика,  нужно  во  время  выбора

очередного перехода набрать (просто на клавиатуре) слово fullvi. Теперь  при  просмотре  инвентаря (по клавише i - как и раньше) будет отображаться  несколько  более  полная информация о текущем состоянии, а именно  значения  всех  переменных  и содержимое инвентаря - как обычно. Такие   сведения   могут быть  очень  полезными  при  отладке  квеста,   содержащего   большое количество   взаимосвязанных   переменных.     

Основы URQL. 

Язык URQL является процедурным, как Pascal, C++, Basic. Файл квеста представляет собой набор операторов, меток и комментариев. 

Работа интерпретатора (важно!)
Интерпретатор начинает обрабатывать файл квеста с самого начала, пропуская комментарии и последовательно читая фрагменты кода оператор за оператором, (Разделителем операторов является символ конца строки или '&') до тех пор, пока не будут встречены операторы

end,

input,

pause,
[new!] anykey

- и в это время не позволяет пользователю работать с переходами на другие локации и инвентарем. Встретив один из операторов остановки, интерпретатор останавливается, и ждет ввода команды пользователя-игрока. Потом выполнение квеста продолжается. Дополнительно, для удобства игрока, запоминается позиция курсора в пройденных локациях. После вывода каждой строчки текста делается задержка заданного количества миллисекунд. Количество это содержится в системной переменной urq_delay (по умолчанию 10, чем меньше, тем быстрее), и может быть заменено пользователем. Преимущества: текст более естественно (и приятно для глаз) появляется на экране, примерно как при прокрутке длинного документа.

Файлы, обрабатываемые движком должны иметь расширением qst или qs1. Файлы qst являются обычными текстовыми документами, а файлы qs1 являются зашифрованной копией файлов qst. Опишем подробнее формат файлов qst: это текстовый скриптовый документ, который выполняется построчно:
1. Вначале отбрасываются комментарии (символы после служебного знака ";"). 

2. Строка разбивается на части служебным знаком "&" и для каждой части выполняются следующие операции. 

Исключение: если встречается оператор "if", то последующая за ним строка на предмет "&" не обрабатывается, эти операторы будут обрабатываться как "then" часть.

3. В выводе информации на экран сочетание символов "#$" заменяется на пробел. Сочетания "#<имя числовой переменной>$" заменяется на их целое значение без дробной части. Сочетание "#%<имя строковой переменной>$" заменяется на текст. Все сочетания символов "#/$" заменяются на перенос строки. 

Примечание: предметы инвентаря прямо не выводятся на экран в позициях #$ и не участвуют в арифметических операциях. При необходимости используйте, доступную в версии 1.3, системную переменную Inv_<Target>.

Комментарии

Комментарий - несколько строк, строка или ее часть, не анализируемая интерпретатором. Используется при разработке квеста для пояснения отдельных фрагментов текста. Существует два типа комметария: 

1. Отделяется при помощи точки с запятой. Все, что правее этого символа, до конца строки - комментарий:

;    Комментарий 

pln Здесь какой-нибудь текст ; А после знака точка с запятой - комментарий.

2. [new!] Можно заключить часть текста между пар символов /* и */ и эта часть будет игнорироваться интерпретатором.

/* Тоже комментарий*/

/* И

  это тоже

 комментарий*/

Метка

Метка - одно из важнейших понятий URQL. Она служит для выделения места в квесте, куда затем предстоит перейти командой вроде btn, goto, proc. Любые перемещения в квесте могут быть либо на следующий оператор, либо на метку. Все метки имеют имена, чтобы потом можно было указывать, куда вы хотите перейти.

Определяется при помощи двоеточия. Все, что правее этого символа, до разделителя, которым является символ конец абзаца (Enter) или '&'  - имя метки:

:Имя_метки

Например, допустимыми именами меток являются 

:1

:01

:а

:кот1

:Название метки с пробелом

:@(a)%^Очень_Длинная_и_Запутанная_метка_какими-лучше-не-пользоваться!!!!!_horror_так_как_сами_потом_забудете_как_она_называется!Или_сделаете_ошибку!Вы_же_можете_сделать_маленькую_ошибку_в таком_длинном_названии?

:house_flat1

Примечание: метка должна начинаться с начала строки и отделяться от других операторов (Enter) или '&', регистр имени метки не важен.

:BIG
:bIG
:Big
:big
 - все эти метки для интерпретатора совершенно одинаковы. Кстати, регистр операторов тоже значения не имеет, можно писать как PLN, так и pln, и Pln и даже pLn - urq_dos - поймет. Но лучше, в целях собственного удобства, придерживаться единого стиля.

В метке лучше не делать пробелов, так как, хотя существующий интерпретатор их поддерживает, это необязательно для следующих версий. Длина метки не ограничена, но для удобства автора квеста лучше не делать меток длиннее строки. В метке допускаются символы, обозначающие буквы русского и латинского алфавита, цифры, знак подчеркивания и знаки арифметических операций + и - (лучше не злоупотреблять странными названиями). Недопустимы символы ';'  '/*', '*/', '&', ','. 

Грамотное использование имен меток способно серьезно облегчить работу автора.

Операторы

"Цепочку команд описать легко, структуру отклика на них - нет."

                                           Урсула ле Гуин,"История шобиков"

Основная часть квеста - операторы, обычно это конструкции вида:

 Имя_оператора параметр_оператора[,другой_параметр]

В качестве параметров можно брать любые строки, но если параметров несколько, то только в последнем можно использовать запятые. Операторы могут быть разделены либо переносом строки, либо символом "&'.

Примечание: регистр операторов не важен, т.е. операторы print, Print, PRINT, pRiNt – выполняют одно и то же действие.
Важнейшие операторы. Создание книги-игры.

Ниже будут описаны наиболее часто используемые в квесте операторы, среди них особенно важны операторы вывода текста, перехода на метку, а также оператор остановки. 

Вывод текста

Самое главное в квесте - вывод текста. Таким образом, в вашем квесте обязательно должен быть, по крайней мере, один оператор P или PLN.

P Текст

После его выполнения будет выведен текст "Текст".

Существуют модификации этого оператора: PLn, Print, PrintLn. Делают они одно и то же, но операторы с Ln на конце делают дополнительно перенос строки. Операторы print и println в настоящее время практически не используются. Возможно, их поддержка будет отключена в следующих версиях досурки.

Переход между локациями

Любой текстовый квест можно разбить на локации с текстом и переходы между ними. Из каждой локации к другим локациям обычно ведет несколько переходов, которые обозначают некоторые альтернативные варианты действий. В URQ переходы осуществляются путем перемещения на указанную метку с помощью операторов Btn и GoTo.

btn Метка, Текст

Примечание 1:  пробелы  до  и  после 'Текста' игнорируются. Возможно,

будет введен очередной флаг - системная переменная для переключения этого.

Примечание 2: Если длина 'Текста' - надписи на кнопке превышает 77 символов, то выводится помещающаяся часть, дополненная символами " >". В следующих версиях в таком случае надпись можно будет скроллировать.

Примечание 3: если Метка не найдена, то кнопка выводится с пометкой // phantom,

отображение кнопок-"фантомов" можно отключить, присвоив системной переменной hide_phantoms значение 1.          (hide_phantoms=1) 

После выполнения оператора

Btn Метка, Текст

интерпретатором создается кнопка - переход на другую локацию. К примеру, в версии urq_dos создается кнопка-строка с текстом "Текст", при нажатии на которую (выбрать ее клавишами курсора вверх-вниз и нажать Enter на клавиатуре), произойдет переход на другую локацию, то есть на метку, носящую имя "Метка". При наличии локации common (см. ниже) сначала выполняются операторы на локации common, потом осуществляется переход на метку "Метка"

GoTo Метка

После выполнения оператора goto метка происходит мгновенное перемещение на указанную метку. При наличии локации common (см. ниже) операторы на локации common при переходе по goto НЕ ВЫПОЛНЯЮТСЯ. Это  сделано специально и умышленно - чтобы можно было делать переходы внутри локации.

Примечание: в случае не нахождения метки, goto просто игнорируется. У Рипа в документации есть  такое правило - если оператор по каким-то причинам обработать не удалось, он игнорируется, как бы считается комментарием.

Остановка

Оператор остановки End приостанавливает работу интерпретатора, чтобы пользователь мог сделать выбор между предложенными ему вариантами действий. Если меткой локация открывается, то оператором остановки завершается.

End 

- после этого оператора интерпретатор выведет все "собранные" кнопки, остановится и будет ждать выбора пользователем перехода к другой локации.

Используя только вышеперечисленные операторы уже можно написать простой квест - книгу-игру, например такой:

Пример 1

:Начало

pln Летний вечер. Смеркается, но фонари зажгутся еще не скоро. Вы идете по улице и вдруг слышите за кустами какую-то возню, короткий возглас "Помо..", оборвавшийся сдавленным стоном и, произносимый шепотом грязный мат...

btn мимо,Пройти мимо, ускорив шаг.

btn вопрос,Спросить: "Эй, кто там!" 

btn герой,Обойти куст и разобраться в чем дело.

btn нормальный_трус,Добежать до телефона-автомата и позвонить в милицию.

end
:мимо

pln Ты благополучно добрался до дома.

btn конец,Конец

btn начало,Еще раз?

end 

:нормальный_трус

pln Ты добрался до телефона, но он как всегда, не работал - кто-то оборвал трубку... Грустный, ты поплелся домой...

goto мимо

:вопрос

pln В ответ слышится громкий грязный мат, обращенный к тебе. По смыслу - тебе советуют для сохранения здоровья отсюда уйти.

btn мимо,Пройти мимо, ускорив шаг.

btn герой,Обойти куст и разобраться в чем дело.

btn нормальный_трус,Добежать до телефона-автомата и позвонить в милицию.

end 

:герой

pln Ты обошел кусты и увидел, как двое каких-то молодых парней прижали к земле девушку и пытаются ее изнасиловать! Она отбивается, но силы явно неравны!

btn мимо,Убежать потихоньку

btn идиот,Крикнуть: "Эй, вы! немедленно отпустите ее!" 

btn идиот,Спросить: "А что это вы тут делаете?"

btn каратэ,С разбегу пнуть того, который поближе.

end 

:каратэ

pln Ловким пинком в голову ты опрокидываешь одного подонка на землю, второй, который лежал сверху на девушке, пытается подняться, но она вовремя хватает его за руку и ты двумя мощными ударами лишаешь его всякой боеспособности! Потом добавляешь обоим для верности еще по пинку, помогаешь девушке подняться и вы убегаете, держась за руку, за угол. Чуть-чуть отдышавшись, вы рассматриваете друг друга: она, несмотря на заплаканное лицо, выглядит очень симпатичной. Ее большие темные глаза полны восхищения. Внезапно улыбнувшись, она высвобождает свою руку из твоей, и удивительно звонким голосом спрашивает: "Как же зовут моего благородного спасителя?" 

btn конец,Представиться

end
:идиот

pln Парни оставили девушку в покое и набросились на тебя! Ты бился как мог, но их было двое, они поднаторели в уличных драках и скоро ты пропустил удар в голову!... Очнулся ты в больнице. Утешало одно - ты поступил как честный человек и девушка, кажется, успела убежать. Жаль, что ты никогда не узнаешь ее имени...

btn конец,А дальше? 

end
:конец

pln Собственно, всё. Это маленький квест. Если есть желание - допишите его сами...

end
Инвентарь. Создание нормального квеста.

Есть прекрасные сложные квесты, состоящие только из локаций и переходов, но, как правило, создатели квестов предпочитают еще и добавлять в игру предметы, заставляя играющих активно работать с ними. При грамотном использовании этой возможности можно значительно разнообразить игровой процесс, есть интересные квесты, которые построены исключительно на манипулировании предметами.

Числовые константы это число вида х[.х] - где каждый х может быть заменен на произвольное количество цифр (Не совсем произвольное, но ограничения сейчас долго искать - длина мантиссы вещественных чисел двойной точности достаточно велика). В отличие от Windows URQ в досурке количество предметов инвентаря может быть и дробным, имейте в виду! Сделано специально, например, ежу понятно, что в кармане может быть 5.67 рублей, то есть 5 рублей 67 копеек.

*Комментарий разработчика:

"Не  помню,  насколько умышленной была эта особенность. И ежу не совсем все  понятно  - если инвентарные переменные там действительно целые, а не  только  выводятся  как  целые  (надо будет посмотреть, так это или нет),  то  это  можно  использовать  для округления чисел, а в Досурке такой   прием   не  сработает.  Возможно,  будет  очередной  флаг  для переключения типа инвентаря."

Inv+ [Количество,]Предмет

После выполнения этого оператора игроку дается Количество (1 по умолчанию) предметов Предмет. Можно прибавить и отрицательное количество предметов - это все равно, что вычесть их из инвентаря. Помните только, что число предматов в инвентаре не может быть меньше нуля - если количество предметов инвентаря после выполнения оператора меньше либо равно нуля, тогда соответствующий предмет из инвентаря убирается.

Inv- [Количество,]Предмет

После выполнения этого оператора у игрока убирается Количество (1 по умолчанию) предметов Предмет. Если количество предметов инвентаря после выполнения оператора меньше либо равно нуля, тогда соответствующий предмет из инвентаря убирается. Убирается  -  это  касается  и inv+. Там просто общая проверка стоит - если  в  результате inv+ или inv- значение стало <=0, то удаляется. На

Количество ограничений и проверок нет.

Проверить наличие (но не количество!) конкретного предмета в инвентаре можно условным оператором if then else, то есть допустимо после оператора 

inv+ 10,Рублей 

проверить 

if Рублей then pln У тебя есть деньги! else pln Так денег-то нет!

Но выражение

if Рублей>5 then pln У тебя больше 5 рублей!

- не работает! Чтобы проверить количество предметов в инвентаре надо или использовать специальную переменную inv_имя предмета, которая равна количеству предметов с названием имя предмета, или старый формат if 5 Рублей then ...,

это  эквивалентно if Inv_Рублей>=5 then ...  Inv_имя   можно   использовать   не   только  для  проверки,  но  и  в арифметических выражениях.

if inv_Рублей>5 then pln У тебя больше 5 рублей!

if 5 Рублей then pln У тебя 5 рублей или больше! 

- обратите внимание на разницу в результате при количестве рублей равном именно 5! Для примера, можно было бы переписать предыдущий маленький квест, вооружив главного героя кирпичом.

Пример 2

:Начало

pln Летний вечер. Смеркается, но фонари зажгутся еще не скоро. Вы идете по улице и вдруг слышите за кустами какую-то возню, короткий возглас "Помо..", оборвавшийся сдавленным стоном и, произносимый шёпотом, грязный мат...

btn мимо,Пройти мимо, ускорив шаг.

btn вопрос,Спросить: "Эй, кто там!" 

btn герой,Обойти куст и разобраться в чем дело.

btn нормальный_трус,Добежать до телефона-автомата и позвонить в милицию.

btn осмотр,Осмотреться вокруг

end
:осмотр

pln Ты оглядываешься вокруг. Как назло, рядом нет прохожих, позвать на помощь некого... Возня за кустами продолжается...

if not Кирпич then pln У самых ног лежит обломок кирпича. 

if not Кирпич then btn кирпич,Схватить кирпич

if Кирпич then btn окно,Разбить кирпичом окно в доме напротив!

btn мимо,Пройти мимо, ускорив шаг.

btn вопрос,Спросить: "Эй, кто там!" 

btn герой,Обойти куст и разобраться в чем дело.

btn нормальный_трус,Добежать до телефона-автомата и позвонить в милицию.

end
:кирпич

pln Кусок кирпича удобно лежит в руке!

inv+ Кирпич

goto осмотр

:окно

pln Ты швыряешь обломок кирпича в окно - раздается серебристый звон разбитого стекла и, через несколько секунд, возмущенный вопль: "Хулиганы! Я милицию вызову!" Возня за кустами прекратилась, затем кто-то крикнул: "Бежим отсюда!" и послышался дробный стук каблуков.

inv- Кирпич

btn ждать,Ждать, что будет дальше

btn обойти_куст,Обойти куст

btn мимо,Убежать

end
:обойти_куст

pln Ты обошел куст и увидел встающую с земли девушку, в разорванной кофточке. Заметив тебя, она видимо приняла тебя за насильника, выхватила из сумочки баллончик и прыснула тебе в лицо! Пока ты протирал слезящиеся от дезодоранта глаза, она врезала тебе чем-то очень твердым между ног и добавила сумочкой по голове! Пока ты загибался от боли, подъехала милиция... Бравые ребята добавили тебе еще несколько раз. В камере предварительного заключения тебя долго выпытывали насчет имен твоих сообщников... Потом отпустили под подписку о невыезде. Потом был суд. И срок. Условно. 

btn конец,А что дальше?

end
:ждать

pln Ты подождал. Вскоре, действительно, подъехала машина ППС, оттуда выскочили два амбала в форме и задержали тебя. "Это он, он стекло разбил!" - голосила внезапно возникшая откуда-то старушка. Тебя задержали за мелкое хулиганство. Стекло пришлось вставить за свой счет.

btn конец,И что?

end
:мимо

pln Ты благополучно добрался до дома.

btn конец,Конец

btn начало,Еще раз?

end 

:нормальный_трус

pln Ты добрался до телефона, но он как всегда, не работал - кто-то оборвал трубку... Грустный, ты поплелся домой...

goto мимо

:вопрос

pln В ответ слышится уже громкий грязный мат, обращенный к тебе. По смыслу - тебе советуют для сохранения здоровья отсюда уйти.

btn мимо,Пройти мимо, ускорив шаг.

btn герой,Обойти куст и разобраться в чем дело.

btn нормальный_трус,Добежать до телефона-автомата и позвонить в милицию.

end 

:герой

pln Ты обошел кусты и увидел, как двое каких-то молодых парней прижали к земле девушку и пытаются ее изнасиловать! Она отбивается, но силы явно неравны!

btn мимо,Убежать потихоньку

btn идиот,Крикнуть: "Эй, вы! немедленно отпустите ее!" 

btn идиот,Спросить: "А что это вы тут делаете?"

if Кирпич then btn удар_кирпичом,Врезать кирпичом по башке тому, который держит девушку.

btn каратэ,С разбегу пнуть того, который поближе.

end 

:удар_кирпичом

pln Ты подбегаешь к негодяю и бьешь его кирпичом по голове! Второй, который лежал сверху на девушке, пытается подняться, но она вовремя хватает его за руку и ты ударом кирпича по уху лишаешь его всякой боеспособности! Потом добавляешь обоим для верности еще по пинку, помогаешь девушке подняться и вы убегаете, держась за руку, за угол. Чуть-чуть отдышавшись, вы рассматриваете друг друга: она, несмотря на заплаканное лицо, выглядит очень симпатичной. Ее большие темные глаза полны восхищения. Внезапно улыбнувшись, она высвобождает свою руку из твоей, и удивительно звонким голосом спрашивает: "Как же зовут моего благородного спасителя?" 

inv- Кирпич

btn конец,Представиться

end
:каратэ

pln Ловким пинком в голову ты опрокидываешь одного подонка на землю, второй, который лежал сверху на девушке, пытается подняться, но она вовремя хватает его за руку и ты двумя мощными ударами лишаешь его всякой боеспособности! Потом добавляешь обоим для верности еще по пинку, помогаешь девушке подняться и вы убегаете, держась за руку, за угол. Чуть-чуть отдышавшись, вы рассматриваете друг друга: она, несмотря на заплаканное лицо, выглядит очень симпатичной. Ее большие темные глаза полны восхищения. Внезапно улыбнувшись, она высвобождает свою руку из твоей, и удивительно звонким голосом спрашивает: "Как же зовут моего благородного спасителя?" 

btn конец,Представиться

end
:идиот

pln Парни оставили девушку в покое и набросились на тебя! Ты бился как мог, но их было двое, они поднаторели в уличных драках и скоро ты пропустил удар в голову!... Очнулся ты в больнице. Утешало одно - ты поступил как честный человек и девушка, кажется, успела убежать. Жаль, что ты никогда не узнаешь ее имени...

btn конец,А дальше? 

end
:конец

pln Собственно, всё. Это маленький квест. Если есть желание - допишите его сами...

end
Расширенная работа с инвентарем

Чтобы упростить доступ к предметам инвентаря и одновременно сделать доступ к ним из выражений есть специальная группа переменных: Inv_предмет1 для работы с инвентарем. Значение переменной Inv_предмет1 равно кол-ву предмет1'ов в вашем инвентаре. Возможна как запись, так и чтение.

Активный инвентарь

Есть возможность организовать обработку действий игрока над предметами. Поэтому введена группа локаций, для обеспечения этого интерфейса.

:Use_Inv - действие Инвентарь/Осмотреть

:Use_Inv_действие1 - действие Инвентарь/действие1

:Use_предмет1 - действие предмет1/Осмотреть

:Use_предмет1_действие1 - действие предмет1/действие1

Примечание: при создании локаций с такими именами действия над предметами добавляются автоматически. Просмотр инвентаря осуществляется по клавише 'I', использование по 'U'. Использование предмета - это подпрограмма из которой идет возврат в ту же локацию, где вы были. Еще  при  возврате  из  I  и  U  на  время  выполнения  одной  локации устанавливается режим только чтение для всех переменных, если бы этого не было и в локации, например, есть что-нибудь вроде inv+ 10,патронов, то  путем  просмотра  инвентаря  можно  было бы накрутить себе гораздо больше боеприпасов. Для примера можете добавить в пример две локации и проверить результат по клавише U.

:Use_кирпич

pln Обыкновенная половинка красного строительного кирпича.

end
:Use_кирпич_выбросить

inv- Кирпич

pln Ты выбрасываешь кирпич прочь. И что дальше?

end
Переменные

Но даже при использовании всех перечисленных выше возможностей иногда возникает проблема - как сохранить то, что игрок где-то был, или создать изменяющийся параметр, который игроку показываться не должен? Это делается с помощью переменных. Переменные, это автоматически создающиеся структуры, которые могут хранить вещественное десятичное число в большом диапазоне: 

- переменные создаются при обращении к ним, если переменная не создана, то при  обращении к ней она равна нулю.

- никакие альтернативные формы записи (2e+5, другие системы счисления ...) пока не поддерживаются.

- имя числовой переменной подчиняется общим требованиям к строке (то есть допустимы пробелы и практически любые символы, кроме '#' '$' ',' ';' '/*' '*/' '&'  но для совместимости со следующими версиями лучше не использовать других знаков, кроме букв русского или латинского алфавита, знака подчеркивания, цифр. Еще переменные  нельзя называть ключевыми словами типа if, then, not, and, or и др. так как в выражениях такие имена будут распознаваться как ключевые слова.

Работать с переменными можно при помощи следующего оператора:

Переменная=Выражение

При этом переменной присваивается значение выражения.  

Выражение это комбинация знаков действий (+ - * /), скобок, переменных и числовых констант. Приоритеты действий соответствуют обычно принимаемым в алгебре.  

Например:

окно_разбито=1

деньги=деньги+взятка

pi=3.1415926

окружность=2*pi*радиус 

сложное_выражение=pi*(радиус*радиус)/2*высота_трубы

[new!]В выражениях полностью поддерживаются операции, в том числе и логические, (в порядке возрастания приоритета):

not, унарные + и - , *, /, +, -, <, <=, [n] предмет, >, >=, = (==), <>, and, or, скобки ( )

Пример 3
pln #5*((2<3)and(3<4))+2$              ;    выдаст 7

Примечание: Условия вида if not a=b ... обрабатываются как if 'not a'=b ... Для записи условия "если не a=b" следует использовать скобки: if not (a=b) ...

Условия

Предметы инвентаря и переменные создаются для активного влияния на ход игры, чтобы реализовать эту возможность существует оператор условия, в двух вариантах:

If Условие then Действия

или

[new!]   if Условие then Действия1 else Действия2

Если выполняется Условие, то производятся Действия.

Условие - комбинация условий предметов ([Количество] Предмет ), условий выражений (Выражение1~Выражение2, где ~ это = > >= < <= <>) и логических операндов (Not And Or). еще  поддерживается  == (оператор неполного совпадения строк), в случае вещественных переменных == пока просто аналог  = (это с мыслью о Сишных = и ==: = присваивание, == сравнение. В  Си,  таким  образом,  можно включать присваивания в выражения, но в Досурке этого пока нет)

Действия - несколько операторов в одной строке, разделенных &.

Пример 4
if Дубина then btn шмяк,Ударить зеленого гоблина дубиной по башке! else btn драп,Драпать!

if Спички and Факел then pln Ты можешь зажечь факел.&btn зажечь_факел,Осветить пещеру.

if not (головная_боль=1 or музыка_звучит=1) then pln Ты наконец-то вспомнил нужный телефон!

Вставка выражений

Часто возникает необходимость, например вывести значение какой-нибудь переменной на экран. Но если просто включить имя переменной после оператора печати p или pln - будет выведено именно это имя, но никак не значение переменной! Для вывода значений был сделан механизм предобработки операторов: если в строке встречаются символы # и $, то они обрабатываются особо. А именно: #Выражение$ в коде заменяется на значение Выражения.

Примеры 5: 
a=1

pln Текст#a$Текст2

Будет напечатано: Текст1Текст2.

if inv_Рублей>70 then inv- 70,Рублей&inv+ Курица&p Ты платишь деньги, продавщица подает тебе замороженную курицу и #inv_Рублей$ рублей сдачи. else pln Курица стоит 70 рублей! У тебя есть только #inv_Рублей$ рублей!  

Примечание: если число не целое, то число выведенных после запятой знаков будет равно значению системной переменной fp_prec; обратите внимание, что конструкция #...$ может быть вставлена не только в оператор PLn, например btn a#a$,Пошли вполне уместно. А также во многие другие места квеста!

Пример 6: Вычисление целой и дробной части числа.

:1

pln Введите число в формате ххх.ххх

input x   ;Вводим произвольное число.

fp_prec=0 ;Задаем число знаков после запятой равное нулю

int_x=#x$ ;Присваиваем переменной int_x целое число.

;Две строчки ниже позволяют сейчас решить проблему с округлением числа вверх.

if x>0 and int_x>x then int_x=int_x-1 ;Для положительных чисел

if x<0 and int_x<x then int_x=int_x+1 ;Для отрицательных чисел

fp_prec=5 ;Возвращаем точность вычислений до 5-го знака после запятой.

frac_x=x-#int_x$ ;Вычисляем дробную часть.

pln Вы ввели #x$ ;Выводим результаты вычислений на печать.

pln Int x=#int_x$

pln Frac x=#frac_x$

btn 1,Еще раз?

btn 2,Хватит

End
:2

end
Звуки и  музыка.

Итак, мы уже можем писать несложные квесты. Но некоторым особо привередливым игрокам хочется, чтобы во время квеста что-то звучало: в пещере капала вода, в бою гремел марш, звенели сабли, и стреляли пушки, а при свидании с любимой девушкой играла "Апассионата". Как это сделать? URQ_dos (пока, - увы, -  только в WCL-версии) поддерживает проигрывание звуков в оператором play
Play имя_файла.расширение_файла

Например:  play intro.mid
Поддерживается не  только  .mid,  так  можно  запустить  файл  любого  медиа-формата, поддержка  которого есть на системном уровне Windows - например, .wav, .mp3  (возможно,  есть  зависимость  от  версии  Windows), .avi (будет загружен  проигрыватель с указанным файлом). Оператор play может проигрывать файлы из других директорий - надо только указать в имени файла - путь. 
Пример 7: Использование звуков и музыки
play sound\intro.wav ; Проигрывает файл с именем intro.wav, лежащий во вложенной по отношению к файлу квеста папке с именем sound.

Многие авторы квестов для экономии места используют именно формат mid (миди-файлы), являющиеся самым компактным способом записи музыки. Их можно проигрывать также при помощи числовой системной переменной music. Она не сможет проиграть mid-файл с именем intro.mid, но зато ей достаточно присвоить не полное имя файла с расширением, а только номер mid-файла. Условие - музыкальный файл должен быть в той же директории, что и файл квеста.

music=1 ; Проигрывает файл с именем 1.mid
После загрузки файла работа квеста продолжается, новый звуковой файл останавливает предыдущий. music=0 - выключает музыку.

Цвет текста, фона и кнопок.

Цвет текста, фона и кнопок зависит от системных переменных:

Style_DOS_FillEol=1 ;эта переменная определяет стиль закрашивания фона цветом, если она равна нулю, закрашивается только фон под буквами текста, если она равна 1, то закрашивание идет по всей длине строки.

Style_DOS_TextColor=7 ; цвет текста и фона за текстом.

Style_DOS_ButtonColor=15 ; цвет кнопок

Style_DOS_CursorColor=112 ;цвет активной кнопки-курсора.

Названия выглядят достаточно "страшно", но они возникли как "наследие" WinURQ от RipOs'a, у которого соответствующие переменные начинались просто Style_* - Виктор Корянов вынужден был ввести свои системные переменные (количество цветов под Windows и под DOS отличаются), но для единообразия добавил префикс DOS после Style: Style_DOS
В примере приведены значения по умолчанию, соответствующие серому тексту на черном фоне, белым кнопкам, инвертированному курсору. Кодировка цвета идет по следующему принципу: в младших 4-х битах цвет, дальше в 3-х фон - это эквивалентно, цвет+16*фон. То есть  Style_DOS_TextColor=7 означает 7+16*0 - серый цвет на черном фоне,Style_DOS_TextColor=31 - 15+16*1 - белый цвет на темно синем фоне. Все цвета можно посмотреть в квесте-примере colors.qst.

Вот фрагмент из этого примера.

:0

Style_DOS_FillEol=0

Style_DOS_TextColor=7

pln == Палитра ==

pln <цвет>+16*<фон>

i=0

:loopi

j=0

:loopj

Style_DOS_TextColor=i+16*j

;p #i$,#j$

if Style_DOS_TextColor<10 then p ##32$

if Style_DOS_TextColor<100 then p ##32$

p ##32$#Style_DOS_TextColor$;

j=j+1

if j<16 then goto loopj

i=i+1

pln

if i<16 then goto loopi

;Style_DOS_ButtonColor=25

Style_DOS_TextColor=7

Style_DOS_ButtonColor=8

Style_DOS_CursorColor=11

btn 0,палитра

end
Процедуры.

Если надо несколько раз повторять одну и ту же последовательность действий, то можно воспользоваться процедурой - оператором Proc имя.

при вызове данного оператора происходит переход на метку :имя и выполнение программы, до встречи первого end, потом управление передается оператору, следующему непосредственно за proc, откуда процедура была вызвана. Допустима  любая  вложенность proc-вызовов (вплоть до использования рекурсии!). Нормально обрабатываются последовательные переходы по proc'у - цепочка ведется до тех пор, пока не встретится первый end. При этом кнопки во всех пройденных локациях накапливаются(!), но при обнаружении end'а происходит не опрос пользователя, а возврат к предыдущему proc'у и так до самого начала цепочки.

Локация common обрабатывается именно так, как должна, то есть как процедура. Это означает, что сняты имевшиеся ранее ограничения на использование переходов в common-локации - теперь из нее, как из процедуры, можно строить сколь угодно длинные цепочки вызовов и они будут правильно обработаны - по окончании цепочки управление вернется в текущую локацию common, а из нее в локацию-метку.

Например, для того, чтобы не писать каждый раз эти "страшные длинные названия" переменных установки цвета, можно использовать их через процедуры:

Пример 9: переключение цвета текста
;----------------Процедуры установки цветов--------------------

:grey
Style_DOS_TextColor=7

end

:darkblue

Style_DOS_TextColor=9

end

:darkgrey

Style_DOS_TextColor=8

end

:green

Style_DOS_TextColor=10

end

:blue

Style_DOS_TextColor=11

end

:red

Style_DOS_TextColor=12

end

:violet

Style_DOS_TextColor=13

end

:yellow

Style_DOS_TextColor=14

end

:white

Style_DOS_TextColor=15

end
;--------Конец процедур установки цвета-----------

А дальше вызывать их в квесте

proc yellow

pln Жетый текст

proc green

pln Зеленый текст

или

proc yellow&pln Желтый текст&proc green&pln Зеленый текст
Вставка значения во время исполнения квеста. 
Предобработка квеста позволяет интерпретатору менять содержание квеста прямо во время выполнения, и эта возможность уникальна, то есть не встречается, насколько мне известно в других языках программирования. Итак:

- Выражение #переменная_x$ заменяется на значение числовой переменной 'переменная_x'(с учетом значения системной переменной fp_prec);

- При обработке строки #$ заменяется на пробел (это  было  сделано,  чтобы  была возможность вставить в текст жесткий пробел, который точно будет выведен); 

- При обработке строки #/$ заменяется на перенос строки, что позволяет обойтись в локации одним оператором печати, даже при печати стихов;

- После оператора печати выражение ##код$ заменяется на символ с этим кодом, где код - число, а не выражение. ##$   работает  пока  только  в  операторах  печати,  недопустима вложенность ##$. Конструкция введена чтобы была возможность включить в текст "системные" символы ;&#$. Код может быть от 0 до 255. Пока нельзя использовать переменную вместо числового значения код!

- Выражение #%строковая_переменная$ заменяется на значение строковой переменной с именем 'строковая_переменная', но об этом более подробно будет рассказано ниже.

Примечание: процесс обработки строки квеста идет следующим образом - слева направо ищется пара # и $, их содержимое обрабатывается и строка обрабатывается снова, до тех пор, пока не кончатся #. Возможно вложение, то есть такая строка считается допустимой: #var#x$$. Кстати, это очень полезная конструкция, см. главу Массивы. 

Строковые переменные и работа с ними

Строковые переменные (<стр> - имя стр. переменной, угловые скобки - не нужны):

создание:

 instr <стр>=<текст>             

вставка значения: 
 #%<стр>$ 

присваивание: пока только <стр1>=<стр2>    т.е. сравнивается только одна строковая переменная с другой!
                   равносильно instr <стр1>=#%<стр2>$

сравнение: пока только 
if <стр1>=<стр2>, но возможно еще

[new!]: 


if <стр1>="текст" - при этом создается строковая переменная содержащая "текст" (без кавычек!) с именем "текст" (без кавычек!)

: if <стр1>=="текст" - оператор неточного сравнения. 

Работает так:

instr k=                    ;обратите внимание на '=' после instr k!!!

pln Напишите в каком направлении должен идти герой:

input k
if k=="*север*" then pln Герой очутился среди белых медведей else pln Куда-куда ты меня послал?

Интересно, что в данном примере, распознаются по условию then фразы "идти на север", "север", "765783север3424юг" и даже "куда угодно, только не на север!"

Возможно и распознавание целой фразы

if k=="*идти*север*" then pln Герой очутился среди белых медведей else pln Куда-куда ты меня послал?

правильно распознает фразы "идти на север","идти в северном направлении" (и неправильно, -увы,- "идти подальше от всех северных медведей!") Кавычки обязательны, так как они в операторе if then служат для формирования из строки текста - текстовой переменной.

Кроме символа '*' заменяющего произвольное количество других символов (в том числе и их отсутствие) возможно применение символа '?', который заменяет ОДИН произвольный символ (или его отсутствие).

instr k= 

pln Напишите ваше любимое оружие

input k
if k=="меч?" then pln Вы получаете волшебный меч! else pln Вы получаете по физиономии!

- правильно распознает и "меч" и "мечи" и правильно ‘не распознает’ "мечту легионера - М-16"

Примечание: в  instr'е  кавычки  пока не обрабатываются, то есть instr v="строка" - так и заносится в переменную v - "строка", с кавычками. 
Оператор разбиения строки - tokens
Для удобства работы с длинными строками введен оператор разбиения строки на отдельные части. Разбиение проводится по разделителям, хранящимся в строковой системной переменной tokens_delim (по умолчанию в ней хранится пробел и четыре знака препинания: точка, запятая, вопрос и восклицальный знак, то есть она разбивает предложение на отдельные слова)

Массивы

Используя #$ и #%$ легко организовать работу с массивами. Массив - это множество переменных, имеющих кроме имени еще и число-индекс, например a1,a2,a3 и т.д. Для двух-, трех-, n-мерных массивов нужно просто 2,3,n индексов разделенных каким-либо разделителем, например a1_1, клетка2|5, кубик4x67y0z. 

Вывод значений таких переменных - элементов массива осуществляется при помощи #$:

кубик4x67y0z=56

x=4

y=67

z=0

pln #кубик#x$x#y$y#z$z$.

- Будет напечатано 67. Если вы поняли, как это работает, можете считать, что с самым сложным моментом в досурке вы разобрались!

Циклы

Присваивать и выводить значения массивам можно в цикле. Циклы также применяются вообще для каких-либо повторяющихся действий. В настоящее время циклы реализованы посредством рекурсии.

i=0

imax=25 ;Инициализация счетчика цикла 

:loop1  ;Метка - начало цикла

i=i+1   ;Увеличение счетчика на 1

if i<imax then p #i$#$&goto loop1 else pln #/$Конец цикла!   ; это "страшное" выражение мы разберем по частям, чтобы было понятно (те, кому понятно и так, могут пропустить это пояснение, учебник пишется для начинающих авторов!)

if i<imax  ;проверяется, что i пока меньше imax, то есть 25

then p #i$#$&goto loop1 ; если да, то печатаем в одну строку числовое значение переменной i (#i$), потом вставляем пробел (#$), потом идет разделитель операторов & и оператор безусловного перехода goto на метку loop1, после которой снова увеличивается i и опять всё по кругу, пока значение переменной i не достигнет 25. Тогда сработает ветка 

else pln #/$Конец цикла! ; которая напечатает перенос строки (#/$) и фразу "Конец цикла!"

Циклы можно вкладывать друг в друга - только применяйте разные названия меток начала циклов, помните, что из двух одинаковых меток в досурке выполняется только первая!

Дополнительный ввод

При обработке квестов иногда надо запросить число. Сделать это стандартными методами - через кнопки, - трудно, поэтому есть оператор

Input x
он добивается от пользователя числового(!) значения и помещает его в переменную 'x'. Если же до ввода определить переменную X как строковую, 

instr X=

input X
- то в переменную X, даже если вы напишите 3.14 будет введено не число 3.14, а строка "3.14". Если же до ввода переменная не определена, то (возм. пока) считается, что вводится число.

Случайность

Иногда полезно - особенно, если не злоупотреблять этим, ввести в квест элемент случайности, непредсказуемости. Именно для этого был введен механизм случайности: Rnd[x] - системная переменная (только для чтения) которая при пустом значении x хранит в себе случайное значение от 0 до 1, включая 1, а при целом x  (к примеру, X=6) выдает выдает целые значения (от 1 до 6  включительно в нашем случае). При каждом обращении к переменной она принимает другое значение. 

случай1=rnd  ;В результате переменная случай1 равна, например, 0.78

случай2=rnd4  ;В результате переменная случай2 равна, например, 3
Время

Иногда для работы квеста требуется системное время, для этого предназначена переменная (только для чтения) Time, она теоретически хранит число секунд прошедших с полуночи. Что она там на самом деле хранит - это вопрос пока сложный и загадочный :) - учитывая разницу в версиях dos32 и wcl. Пока к использованию не рекомендуется.

Многофайловые квесты

Есть возможность создавать квесты состоящие из нескольких qst файлов, для этого 

используется оператор include:

include путь_к_файлу_"квест_вставка.qst" ;имя файла надо брать в кавычки!

который вставит в ваш квест в это место квест_вставка.qst Пока что, следует быть внимательным с использованием вставок - во вставленных квестах, пока не работает вставка значения во время выполнения через #$.

Операторы задержки выполнения интерпретатора

Существует два очень полезных оператора, способных сделать ваш квест по-настоящему интерактивным: pause [время] и anykey [переменная]

pause <время> - прекращение выполнения инструкций на <время> в миллисекундах (1 миллисекунда=1/1000 доли секунды), причем пользователь в это время может выбирать варианты с помощью кнопок выбора и квест будет работать дальше, но по истечении <время> выполнение квеста продолжиться, одновременно двух работающих "pause" не может быть, новая затирает старую - движок о ней "забывает", забывает также он о паузе при нажатии кнопки действия. 

:1

pln Сигнал тревоги разнесся по коридору. Надо спешить, скоро здесь будет охрана.

btn LeftDoor,левая дверь

btn RightDoor,правая дверь

pause 3000 ;Пауза на три секунды
goto LeftDoor

end

:LeftDoor

pln Поздравляем! Вы выскочили прямо навстречу бегущей охране...

end
:RightDoor
pln Спасены! Охрана пробегает мимо.

save
end 

anykey [числовая_переменная]

Этот оператор останавливает выполнение квеста до нажатия любой клавиши на клавиатуре, значение ASCII кода клавиши заносится в переменную "числовая_переменная" и может быть использовано впоследствии. При помощи этого оператора на досурке можно реализовать даже простейшую текстовую аркаду!

Вход на локацию

Очень интересной особенностью является то, что при входе на локацию после перехода (Btn, Goto, Proc) происходит увеличение переменной с именем count_имя-локации на 1. Пример:

proc abs & proc abs & proc abs

end

:abs

pln #count_abs$

end
При этом после выполнения кода (интересно, что впримере 1ый end можно не писать) переменная count_abs примет значение 3 (если до этого не вызывалась abs). Учитывая, что после возврата из просмотра инвентаря или работы с действиями count_имя локации увеличивается также, не рекомендуется применять переменные-счетчики для действительного определения числа посещений локации - лучше использовать обычные переменные.

Вот некоторые вопросы, остающиеся после просмотра учебника: 

Q: Что будет, если сделать ссылку на несуществующую метку? 

A: В досурке кнопка-переход на несуществующую метку либо не действует, помечаясь при этом словом phantom (если системная переменная Hide_phantom=0), либо вообще невидимы (если системная переменная Hide_phantom=1), goto и proc на несуществующую метку просто не выполняется.

Q: Что будет, если создать несколько меток с одинаковым именем?

A: Сделать их можно, но переход будет происходить только на первую из них.

Q: Рассказать о метках Common, упомянуть переменную common, пояснить, что будет, если присвоить переменной нулевое или отрицательное значение, а также значение, для которого нет соответствующей метки Common_Номер. Что будет, если создать метку с именем Common_-128, когда их всего две (Common и Common_-128), то есть, во-первых, корректно ли такое имя, а во-вторых, возможно ли создавать метки вида Common, перескакивая через номера (Common, Common_5, Common_12, Common23), или следует создавать их в строгой последовательности (Common, Common_1, Common_2, Common_3 и так далее)? Что будет, если создать несколько меток Common с одинаковым номером? Неплохо бы также упомянуть альтернативу локациям Common и переменной common: можно ведь создать переменную НомерМетки, присвоив ей значение 0, несколько меток (скажем, Метка_0 Метка_1, Метка_2, Метка_3 и так далее), а потом в начале каждой локации писать строчку "proc Метка_#НомерМетки$" (без кавычек, разумеется). Пояснить, как RipOs и Корянов облагодетельствовали всех квестописателей, избавив их от этой нудной работы :) 

A: Подробнее о Common метках:

Часто необходимо выполнять некую общую последовательность действий для всех локаций... Именно для этого есть 

:Common метка - при вызове btn метка,Название метки на самом делается следующее: proc Common & GoTo метка к примеру, то есть вызывается Common и при встрече end идет выход из процедуры и только после этого уже переход на требуемую метку.

Но ведь для разных частей квеста порой требуется разные Common локации, поэтому системная числовая переменная Common отвечает за имя вызываемой локации: :Common_номер к примеру при Common=5 будет вызов :Common_5. Никаких ограничений по номерам (подряд или нет) не накладывается, для простоты рекомендуется использовать только целые числа. Если Common=-5, то вызов будет соотвественно :Common_-5.

Можно также обойтись и без использования локации Common вообще, например в КАЖДОЙ локации писать proc CommonLoc_#CommonLoc_N$, при этом будет происходить подобный же механизм(только имя Common локации будет :CommonLoc). Использование локации common сильно упрощает написание квестов, в которых надо вести учет ходов игрока (например, для введения игрового времени, голода, запаса энергии в лампе и т.п.)

Q: Как нормально сделать голод в квесте? 

A: Используйте локацию Common, например, так:

:Common 

Сытость=Сытость-1 

if Сытость<=0 then PrintLn Вы голодны!!! 

if Хлеб then Сытость=Сытость+10 & Inv- Хлеб 

end

Q: Равнозначны ли k=5 и k = 5 ? То есть пробелы как-то убираются?
A: При вещественных присваиваниях – да, убираются. При строковых (если k объявлена через instr k= как строковая переменная) можно включить сохранение пробелов через системную переменную Instr_leave_spc - если отлична от нуля, то в instr-присваиваниях сохраняются пробелы, иначе нет.
Q: Пишу instr text=Пароль
If text=Пароль then … - и не работает! Почему?
А: if сравнение чего-то=текст может быть только if чего-то="текст", так как сравнивается только содержимое текстовых переменных, а в последнем случае как раз создается переменная с именем и содержанием text.
Q:  Пишу сравнение двух строковых переменных if stroka1<>stroka2 then … - но работает неправильно!
A: Правильно писать  if not (stroka1=stroka2) then … так как неравенство для строковых переменных не поддерживается!
Q: Как узнать длину строковой переменной?

А:. Длина строки заносится в числовую переменную с тем же именем! Таким образом если instr слово=Пароль, то Длина_слово=#слово$ 

Кстати, при сравнении строки с числом сравнивается именно ее длина.
Q: И еще вопрос: есть ли в УРК команды для работы со строками? Т.е. ты input строчку, а потом, скажем, рассматриваешь ее n-ный символ (парсер хотел сделать)? Или все придется делать способом дровосека?: 

:main 

i=0 

:cycle 

i=i+1 

anykey l#i$ 

a=l#i$ 

if a=13 then goto vvod 

if a=97 or a=65 then p ф 

if a=98 or a=66 then p и 

if a=99 or a=67 then p с 

... 

goto cycle 

end 

А(Walky Talky):
Это можно сделать при помощи операции расщепления на подстроки (токены). 

tokens_delim - строковая переменная, по которой делится строка. Если равна char, то строка делится на отдельные символы. 

tokens - команда делит строку на подстроки (разделитель - tokens_delim) Результаты сохраняются в строках вида tokenN, где N - номер подстроки. (очевидно, что если tokens_delim равно char, то в этих строках содержатся отдельные символы) 

tokens_num - содержит число подстрок в результате расщепления строки. 

Операция разделения на подстроки никак не сказывается на исходной текстовой переменной. 

Пример: 

:1 

instr string= 

pln Введите строку: 

input string 

instr tokens_delim=char 

pln Введите позицию символа: 

input n 

tokens string 

if tokens_num>=n then pln #%token#n$$ - #n$-й символ строки "#%string$" else pln Такого символа нет. 

pln #/$В строке #tokens_num$ символов. 

btn 1,еще 

end
Q: Фиксирует ли anykey код нажимаемой кнопки или для этого используется другая команда?

A(Victor): Да, фиксирует. Простой пример: 

:0 

anykey x 

p #x$ ; 

if x<>27 then goto 0 

выводятся коды нажатых клавиш, пока не нажмешь Esc.

Полезные алгоритмы URQL для написания квестов от «мэтров»:
Etev
я теряюсь... вы наверное и сами всё знаете
ну, из последнего, что я сам реализовывал и запомнилось: «как сделать подобие ссылки» 

HealthEnemy1=30;жизнь первого врага 
HealthEnemy2=30;жизнь второго врага 
instr HealthTarget=.;якобы ссылка 

потом запрашиваете у юзера (или через цикл) порядковый номер врага, здесь описывать не буду - элементарно (например input target),  и присваиваете

instr HealthTarget=HealthEnemy#target$ 

а затем можно оперировать напрямую через ссылку в любом выражении, например: 

if toHitHero>armorEnemy then #%HealthTarget$=#%HealthTarget$-5 


Victor

Распечатка токенов: 
:0 
instr a= 
input a 
tokens a 
pln #tokens_num$ tokens: 
i=1 
:loop 
if i‹=tokens_num then pln #i$: #%token#i$$ & i=i+1 & goto loop else ; 
_if not(a=«bye») then goto 0 else end 

Поиск в токенах слова «гоблин»: 
:1 
instr pars=. 
input pars 
tokens pars 
i=1 
:loop 
if token#i$=«гоблин» then pln Я не гоблин! - истерически кричит урук-хай & goto 1 
if token#i$=«пока» then quit 
if i‹tokens_num then i=i+1 & goto loop 
pln Не понимаю 
goto 1 

Пример из 2kw, с которого все началось: 
:0 
instr a= 
input a 
if a==«*проси*гоблин*золот*» then pln гоблин: золота нет else ; 
_if a=«bye» then quit else pln гоблин ничего не ответил 
goto 0
Евгений
данный пример делает кое-что с введенной строкой.
instr_leave_spc=1 
instr k= 
input k 
instr tokens_delim=char [image: image1]
tokens k 
i=0 
:1 
i=i+1 
instr z=#%z$#%token#i$$ 
if i‹=tokens_num then goto 1 
pln #%z$

Этот кусочек кода изменяет цвет всего фона на(в данном случае) синий. 

clrscr=1 
Style_DOS_TextColor=27 
Style_DOS_FillEol=1 
pln ##0$ 
cls
Эффект забавен тем, что выводит подобие сосулек или чего-то подобного по рнд... Код сумбурный, недоработанный, но может кому пригодится :) 

:0 
old_style_dos_textcolor=style_dos_textcolor 
style_dos_textcolor=15 
proc эффект6/*применение эффекта*/ 
style_dos_textcolor=old_style_dos_textcolor 
end 

:эффект6 
clrscr=4096 
instr_leave_spc=1 
z=78 
i=1 
:эффект6_loop0 
b#i$=i 
if i‹z then i=i+1 & goto эффект6_loop0 
i=1 
:эффект6_loop1 
a#i$=#rnd#z$$ 
if b#a#i$$=0 then goto эффект6_loop1 else b#a#i$$=0 
if i‹z then i=i+1 & goto эффект6_loop1 
i=1 
:эффект6_loop2 
simb#a#i$$=1 
simb#a#i+1$$=1&simb#a#i+2$$=1&simb#a#i+3$$=1 
j=1&instr pl=; 
:эффект6_loop3 
if simb#j$=0 then instr pl=#%pl$##15$else instr pl=#%pl$ ; 
if j‹z then j=j+1&goto эффект6_loop3 
pln #%pl$ 
if i‹z then i=i+4 & goto эффект6_loop2 
;if i‹z then i=i+1 & cls& goto эффект6_loop2 
end
Korwin
Для начала предлагаю свою процедуру выведения времени с часами и минутами: 

:common 
m=m+rnd30 ;Каждый ход прибавляется случайное количество минут от одной до 30 
proc minute ;Подсчитывается прибавка минут и часов 
if h›23 then h=h-24&d=d+1 ;Подсчитывается прибавка часов и дней 
p #d$-й день. Время #h$:&if m‹10 then p 0&p #m$.#/$ else p #m$.#/$ ;Осуществляется вывод дней, часов и минут. Обратите внимание - корректно выводится количество минут, меньшее 10! Аналогично можно сделать и с часами, если нужно делать вид «электронных часов». 
end 

:minute 
if m›59 then h=h+1&m=m-60 
if m›59 then goto minute ;Если прибавили более 60 минут, то нужна проверка. 
end
Переключение цветов. 

Чтобы не писать каждый раз: Style_Dos_TextColor=11 (лиловый, к примеру) 
можно создать строковую переменную в которую ввести весь этот текст и вставлять ее вызов по мере необходимости в нужное место. 
:Инициализация переменных квеста 
... 
instr vio=Style_Dos_TextColor=11 ;Лиловый 
instr yel=Style_Dos_TextColor=14 ;Желтый 
... 

:Локация желтого цвета 
#%yel$ 
p Этот текст будет желтым &#%vio$&pln Этот текст будет лиловым 
...
как сделать goto в локацию, название которой только что инпут? 
instr loc= 
input loc 
goto #%loc$ 

WalkyTalky
Пример получения значений косинуса и синуса угла (в градусах). 
(только для URQ_DOS): 

:1 
perkill 
fp_prec=4 
prec=0.00001 
inval=0 
pi=3.1415926 
p Введите угол: 
input inval 
smval=#inval$ 
:too_big 
if smval>=360 then smval=#smval-360$ & goto too_big 
if smval<=-360 then smval=#smval+360$ & goto too_big 
proc sinus 
pln sin #inval$##248$ = #retval_sinus$ 
proc cosinus 
pln cos #inval$##248$ = #retval_cosinus$ 
btn 1,Еще 
end 

:sinus 
x=smval 
x=x*pi/180 
n=1 
u=x 
s=x 
if u>=0 then abs=#u$ else abs=#-u$ 
:sinus2 
if abs>prec then proc trigonom_loop & goto sinus2 
retval_sinus=s 
end 

:cosinus 
x=smval 
x=x*pi/180 
n=0 
u=1 
s=1 
if u>=0 then abs=#u$ else abs=#-u$ 
:cosinus2 
if abs>prec then proc trigonom_loop & goto cosinus2 
retval_cosinus=s 
end 

:trigonom_loop 
n=n+2 
u=-u*x*x/((n-1)*n) 
s=s+u 
if u>=0 then abs=#u$ else abs=#-u$ 
end 

Goraph

Паузы меньше секунды в урке, как известно, не работают :) Однако есть алгоритм, который позволяет их реализовывать. 

В начале квеста замеряем величину паузы длиной в секунду (лучше взять паузу побольше если вам нужна точность - пауза в 40-60 секунда обеспечит вам лучшую точность чем секундная пауза). Зачем это нужно? Просто досурка и гуи досурка возвращают разные значение времени, плюс оно может быть разным на разных компьютерах - поэтому мы вычилсляем значение для той среды, в которой мы запущены. 

t1=time 
pause 1000 
t2=time 
t_etalon=(t2-t1)/1000 


;Там где нам нужна пауза, пишем такой код - величина паузы в миллисекундах - в данном случае полсекунды 
пауза=500 & proc stop 


; а вот и сама процедура, которая делает нашу паузу длинной меньше секунды 
:stop 
стоп_таймер=time 
пауза=пауза*t_etalon 
if t_etalon=0 then пауза=1 
:stop_loop 
if (time-стоп_таймер)‹пауза then goto stop_loop 
end 

Данный код придуман и разработан всеми любимым Евгом. Я его лишь несколько уточнил (на этапе временного эталона).
VampirE

Вот еще удобная вещь: 

:pk 

style_dos_textcolor = style_dos_cursorcolor 

p #/$ [ НАЖМИТЕ ПРОБЕЛ ДЛЯ ПРОДОЛЖЕНИЯ ] 

style_dos_textcolor = 15 

:ak 

anykey key 

if key = 27 then pln #/$&quit 

if key <> 32 then goto ak 

cls 

end 

При нажатии пробела происходит очистка экрана и код выполняется дальше. Код цветов можно опустить. Таким образом, где нужно, что бы пользователь нажал кнопку пишем proc pk и все. ;)

Справочник по операторам и системным переменным URQ_DOS версии 1.35

Составитель - Walky Talky, дополнения Korwin.

Внимание: символы <.[...].> написаны для более легкого понимания текста и файле их нет, причем символы [ ] используются для обозначения текста, который можно опустить. Пример: 

  print <строка>      - напечатает просто         строка

Внимание: параметры операторов разделены запятыми, то есть в качестве параметра берется строка от пробела после оператора до запятой, не включительно, а в качестве второго параметра от запятой до конца, кроме последнего пробела. Исключение: "if" имеет особый формат. Пример: 

  btn странная пещера, Войти в странную пещеру, затаиться, и прислушаться.
0) /* Это длинный 

многострочный 

комментарий*/
; Это комментарий в одну строку

_ в начале строки означает, что она является продолжением предыдущей.  Перед 
  _ допустимы пробелы и табуляции.
1). ":" - символ начала локации. * Регистр переменных, предметов, меток и операторов никакой роли не играет. Но мы вам не рекомендуем использовать переменные и предметы инвентаря с пробелами. 

    Использование: :имя_локации

    Пример: :локация

    Примечание: Регистр переменных, предметов, меток и операторов никакой роли не играет. Но мы вам не рекомендуем использовать переменные и предметы инвентаря с пробелами. 

2). "end" - символ конца локации.

    Использование: так и используется, на отдельной строке.

    Пример: -

3). "p" - печатает указанную строку на экране.

    Использование: p строка

    Пример: p Напечатанная строка

    Примечание: Для оператора p есть аналог -  оператор print.

4). "pln" - печатает на экране текст, после него

    всегда (даже если строка и пустая) ставит

    символ перевода строки.

    Пример: pln Програмист - узник замка If.

    Примечание: Для оператора pln существует аналог -  println

Исключение: символы "#<выражение>$" заменяются на значение этого выражения 
Пример: 

p У тебя #money+10$ монет напечатает "У тебя 110 монет", если money=100.

Примечание: если длина введенного текста превышает размер страницы, то вывод приостанавливатеся, появляется надпись «дальше» и интерпретатор ожидает нажатия любой клавиши.
5). "btn <локация>,<надпись>" - создает кнопку для выбора вариантов действия с надписью на ней <надпись>, при клике на ней будет осуществлен переход на локацию "Common" (если такая есть), потом (когда встретиться "end") на <локация>.

    Использование: btn имя локации,надпись на кнопке

    Пример: btn tower,Взобраться на башню

    Примечание: для URQ_DOS существует ограничитель длины надписи на кнопке - 79 символов. В имени локации и названии кнопки допустимо применение #$, #числовая_переменная$, #%строковая_переменная$,  ##код$.
6). "inv+" Добавляет в инвентарь предмет(ы).

    Использование: inv+ [кол-во], Название предмет[a](ов)

    Пример: inv+ 5,Сапог

    Примечание: Не обязательно указывать количество предметов для добавления - если не указано, добавляется 1 предмет. Проверяется, что кол-во предметов положительно, то есть если было 15 Рип и 

"inv+ -20,Рип" то Рип исчезнет из инвентаря.

7). "inv-" Удалаяет предмет(ы) из инветаря.

    Использование: inv- [кол-во], Название предмет[a](ов)

    Примечание: Не обязательно указывать количество предметов для добавления - если не указано, добавляется 1 предмет. Проверяется, что кол-во предметов положительно, то есть если было 15 Рип и "inv-20,Рип" то Рип исчезнет из инвентаря.

8). "inv_" Устанавливает и возвращает количество предметов с заданным названием в инвентаре.

    Использование: inv_НазваниеПредметов

    Пример: 
 slonov=inv_Слонов

           

 inv_Слонов=5

9). "invkill" – без параметров удаляет все предметы из инвентаря.
       Invkill предмет - удаляет «предмет» из инвентаря.
    Удобно использовать в паре с оператором perkill в  начале квеста для очищения памяти.

10). "perkill"  - уничтожает все переменные.

     Примечание: Используется без параметров. Удобно использовать в паре с оператором invkill в начале квеста для очищения памяти.

11). "if условие then оператор&оператор else оператор&оператор" - условная конструкция.

     Использование: if условие then операторы или 

     if условие then операторы else операторы
     Пример: if inv_Спичек=50 then btn buildHouse, Построить домик из спичек

     Примечание 1: Можно делать сложную конструкцию, используя &, and, or, not и скобки. Допустимость любой степени вложенности if'ов сохраняется, else всегда соответствует последнему незакрытому then'у.
12). "proc <метка> "- передает управление на метку <метка>, то есть инструкции квеста выполняются с метки <метка>, но когда встречается "end" возвращает управление обратно, оператору, следующему за proc, при условии, что не было оператора forget_procs. (forget_procs - "забывает" все переходы по процедурам.) Допустима вложенность.
     Использование: proc имя_локации

     Пример: proc tower
     Примечание: при proc  - common не выполняется
13). "goto" - переходит на указанную локацию, но не возвращается обратно. 
     Использование: goto имя_локации

     Пример: goto tower
     Примечание: В отличие от WinURQ, в urq_dos - не выполняется заход на локацию common. При отсутствии метки <метка> - goto игнорируется.

14). "save" - оператор, сохраняющий в файл вида  имя_квеста.qsv текущие значения переменных 

     и предметы инвентаря, чтобы потом можно было этот квест с сохраненного места продолжить.

     Использование: save локация_для_восстановления

     Пример: save tower

15). "input" - просит пользователя ввести строку, значение которой будет потом присвоено указанной

переменной. При необходимости ввести с клавиатуры русский текст через оператор input в строковую переменную ее следует предварительно объявить выражением:
instr строка= ; Данное выражение создает строковую переменную нулевой длины.

Input строка

     Использование input для ввода числовой переменной: 
input переменная_для_записи_числа
     Пример: input x
16). "instr" – создает текстовую переменную и присваивает ей значение.

     Использование: instr имя_переменной=значение

     Пример: 
Instr time=время - 17:42



Instr pusto=                         ;В этом случае значение текстовой переменной равно пустой строке.
17). "#переменная$" - вставляет значение числовой переменной.

     Использование: Обычно так и используется.

     Пример: x=5

             pln #x$

             goto b#x$

             btn b#x$,Неизвестная локация

18). #%переменная$ - вставляет значение строковой переменой.

     Использование: Обычно так и используется.

     Пример: instr x=loc

             pln #%x$

             goto b#%x$

             btn b#%x$,Неизвестная локация

19). "pause" - оставливает выполнение квеста на указанное в операторе количество милисекунд.

     Использование: pause <время> - прекращение выполнения инструкций на <время> в милисекундах, причем пользователь в это время может выбирать варианты с помощью кнопок выбора и квест будет работать дальше, но по истечении <время> выполнение квеста продолжиться, одновременно двух работающих "pause" не может быть, новая затирает старую - движок о ней "забывает", забывает также он о паузе при нажатии кнопки действия. 

     Пример: pause 1000

     Примечание: В URQ_DOS интервал задержек пока не может быть меньше 1 секунды (связано с BIOS).

     В URQ ограничений нет.

20). "music" - проигрывает циклично .mid файл.

     Прогрывает только те файлы, в которых имя состоит лишь из цифр, например – 65.

     Использование: music имя_файла_в_директории_квеста

     Пример: music 17

     Примечание: Конструкция не работает в URQ_DOS. (Не wcl!)
21). "play" - проигрывет однократно файл в формате .wav

     В URQ gроигрывает только те файлы, в которых имя состоит лишь из цифр, например - 580.  
     В URQ_WCL – и файлы с именем в виде строки.


     Использование: play имя_файла_в_директории_квеста

     Пример: play 2051

     Примечание: Конструкция не работает в URQ_DOS. (Не wcl!)

22). "rnd" - генерирует случайное число от 0 до 1, не

     включая 1, т.е. от 0 до 0(9). Rnd7 дает целые случайные числа от 1 до 7, Rnd1245 – от 1 до 1245
     Использование: Можно присвоить переменной, можно вывести в текст, создать кнопку, переход и т.п.

     Пример: x=rnd

             pln #rnd$

             btn #rnd$,Случайная локация.

     Примечание: Можно организовать диапазон не до 0(9), а, например, до 10: x=rnd*10. Допустимо использовать Rnd#числовая_переменная$
23). "cls" - очищает экран ото всяческой информации.

     Использование: Так и используется.

     Примечание: В URQ_DOS есть переменная clrscr, которая

     может принимать значения 0 и 1. Если 0 - то очищается

     только тест текущей локации, если 1 - то очищается 

     ВСЕ на экране.

     Пример: clrscr=1

             cls

24). "&" - конкатекация, aka объединение команд.

     Использование: команда1 & команда2 ... & командаN

     Пример: pause=1000 & pln Текст & clrscr=0

25). "and" - логическое "и".

     Использование: В основном, в условных конструкциях.

     Пример: If inv_Slon=5 and inv_Mouse=2 then btn scare, Шугануть слонов мышами
26). "or" - логическое "или".

     Использование: В основном, в условных конструкциях.

     Пример: if Inv_Slon=1 or inv_Mouse=1 then btn zoo,Да у тебя тут зоопарк!

27). "not" - логическое "нет".

     Использование: В основном, в условных конструкциях.

     Пример: if not Coca-Cola then btn coca-cola,Купить 

             "Кока-Колу"

28). "count_" - подсчет заходов на указанную локацию.

     Использование: count_имя локации

     Пример: x=count_tower

29) "quit" - оператор немедленного выхода из квеста.

28) "tokens" - разложение строки на «токены», то есть куски текста, отделенные друг от друга разделителями. 
Применение: tokens <стр. перем>

Результат: числовая переменная  tokens_num содержит количество токенов (например, слов в предложении), строковые переменные token1, token2 ... token# tokens_num$ содержат сами токены (например, слова). См.     стр. сист. перем. tokens_delim
29) "anykey <переменная>" – оператор приостанавливает работу интерпретатора до нажатия произвольной клавиши, заносит код клавиши в переменную.
30) "include "имя файла" - до выполнения в код вместо этой строки вставляется содержимое указанного файла. Позволяет создавать многофайловые квесты.

31) "==" (два знака равно) – оператор неточного сравнения. Дает истину при сравнении строковой переменной со строкой по правилам сравнения регулярных выражений, т.е. с учетом * и ?
Например: 
Instr фраза=

Input фраза

If фраза=="*парол*" then p Проходи! else p Стоять, руки вверх!
Введенные фразы «Пароль!», «Я знаю пароль!», «Никто не знает пароль» - дадут ответ «Проходи!»
Системные переменные:

Rnd - вещественное случайное число от 0 до 1, меняется при каждом вызове.

Rnd[Num] - целое случайное число от 1 до Num, меняется при каждом вызове.

Music - имя фоновой музыки вроде "104.mid" причем проигрываются только Midi файлы.

Common - номер действующей локации Common, если переменная Common=12, то вызывается по btn-переходу Common_12. При Common=0 вызов вызывается просто "Common" (то есть вместо локации :Common используется :Common_номер)

Instr_leave_spc - если отлична от нуля, то в instr-присваиваниях сохраняются пробелы, иначе нет
Urq_delay -  та самая задержка при выводе строк. Напомню, что после вывода каждой строки на экран теперь делается задержка некоторого количества миллисекунд. По умолчанию это количество равно 10,                 а переопределить его можно, установив URQ_DELAY [пример: set URQ_DELAY=0]. Если [как в этом примере] установить значение 0, то вывод текста будет происходить как раньше.
Urq_coding  если установить значение DOS [SET URQ_CODING=DOS], то кодировка квестов будет по умолчанию DOS'овской
Style_dos_textcolor  - цвет текста, по умолчанию 7
Style_dos_cursorcolor – цвет курсора, по умолчанию 112 (инвертированный)
Style_dos_buttoncolor – цвет «кнопок», по умолчанию 15
Style_DOS_FillEol - ха
current_loc и previous_loc  (строковые системные переменные) хранят имена соответственно текущей и предыдущей посещенной локаций. previous_loc удобно использовать для сохранения игры из любой локации при помощи действия типа use_inv_Запись. Для первой локации previous_loc равна ее имени (чтобы сохранение работало и для нее). current_loc и previous_loc  теперь меняются только при btn-переходах.
Clrscr=0: cls работает как прежде, очищая текст одной локации; 
Clrscr=1: cls очищает весь экран, переводя курсор в левый верхний угол; 

Clrscr=4096 только переводит курсор, без очистки экрана – полезно для текстовой анимации.
hide_phantoms=0 – показывает кнопки ведущие в «никуда» с пометкой «phantom», hide_phantoms=1 – не показывает такие кнопки.
last_btn_caption – строковая переменная, которая содержит название последней выбранной кнопки.
fp_prec – числовая переменная, которая содержит точность вычислений, число знаков после запятой в #$ и инвентаре. По умолчанию 2.
tokens_delim – строковая переменная, содержащая список символов, являющихся разделителями (по умолчанию пробел, запятая, кавычки, вопрос, восклицательный знак. Если в tokens_delim строка char (возможно, будет изменена), то tokens s разбивает s на отдельные символы.
Если кто-то заметит неточности или пожелает внести добавления в этот документ, прошу сообщить мне на tightbow@yandex.ru
С уважением, Korwin.
